# Maintainer: Firmicus <firmicus āt gmx dōt net>
# Maintainer: Rémy Oudompheng <remy@archlinux.org>

pkgname=texlive-langcyrillic
pkgver=2021.58426
_revnr=${pkgver#2021.}
pkgrel=1
pkgdesc="TeX Live - Fonts and macro packages to typeset Cyrillic texts"
license=('GPL')
arch=(any)
depends=('texlive-core')
groups=('texlive-lang')
url='http://tug.org/texlive/'
source=("https://sources.archlinux.org/other/texlive/$pkgname-$pkgver-src.zip" "$pkgname.maps")
options=('!emptydirs')
sha256sums=('9dbab28a29c74164f336f38d735277b46ae837492a6fe417b96345620979db7f'
            '4ca6db61e59bcb0f718a1cb062dad04e7f01866c5a7d320dcf67e6a47130f222')

build() {
   cd "$srcdir"
   for p in *.tar.xz; do
	   bsdtar -xf $p
   done
   rm -rf {tlpkg,doc,source} || true
}

package() {
   cd "$srcdir"
   install -m755 -d "$pkgdir/var/lib/texmf/arch/installedpkgs"
   sed -i '/^#/d' CONTENTS
   install -m644 CONTENTS "$pkgdir/var/lib/texmf/arch/installedpkgs/${pkgname}_${_revnr}.pkgs"
   install -m644 $pkgname.maps "$pkgdir/var/lib/texmf/arch/installedpkgs/"
   install -m755 -d "$pkgdir/usr/share"
   wanteddirs=$(for d in *; do test -d $d && [[ $d != texmf* ]] && echo $d; done) || true
   for dir in $wanteddirs; do
     find $dir -type d -exec install -d -m755 "$pkgdir"/usr/share/texmf-dist/'{}' \;
     find $dir -type f -exec install -m644 '{}' "$pkgdir"/usr/share/texmf-dist/'{}' \;
   done
   if [[ -d texmf-dist ]]; then
     find texmf-dist -type d -exec install -d -m755 "$pkgdir"/usr/share/'{}' \;
     find texmf-dist -type f -exec install -m644 '{}' "$pkgdir"/usr/share/'{}' \;
   fi
   if [[ -d "$pkgdir/usr/share/texmf-dist/scripts" ]]; then
     find "$pkgdir/usr/share/texmf-dist/scripts" -type f -exec chmod a+x '{}' \;
   fi
   mkdir -p "${pkgdir}/usr/bin"
   ln -s /usr/share/texmf-dist/scripts/texlive-extra/rubibtex.sh "${pkgdir}/usr/bin/rubibtex"
   ln -s /usr/share/texmf-dist/scripts/texlive-extra/rumakeindex.sh "${pkgdir}/usr/bin/rumakeindex"
}
